E02AJF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

E02AJF determines the coefficients in the Chebyshev-series representation of the indefinite integral of a polynomial given in Chebyshev-series form.

2 Specification

```
 SUBROUTINE E02AJF(NP1, XMIN, XMAX, A, IA1, LA, QATM1, AINT,
1                  IAINT1, LAINT, IFAIL)
 INTEGER           NP1, IA1, LA, IAINT1, LAINT, IFAIL
 real              XMIN, XMAX, A(LA), QATM1, AINT(LAINT)
```

3 Description

This routine forms the polynomial which is the indefinite integral of a given polynomial. Both the original polynomial and its integral are represented in Chebyshev-series form. If supplied with the coefficients a_i, for $i = 0, 1, \ldots, n$, of a polynomial $p(x)$ of degree n, where

$$p(x) = \frac{1}{2}a_0 + a_1 T_1(\bar{x}) + \ldots + a_n T_n(\bar{x}),$$

the routine returns the coefficients a'_i, for $i = 0, 1, \ldots, n+1$, of the polynomial $q(x)$ of degree $n+1$, where

$$q(x) = \frac{1}{2}a'_0 + a'_1 T_1(\bar{x}) + \ldots + a'_{n+1} T_{n+1}(\bar{x}),$$

and

$$q(x) = \int p(x)\, dx.$$

Here $T_j(\bar{x})$ denotes the Chebyshev polynomial of the first kind of degree j with argument $\bar{x}$. It is assumed that the normalised variable $\bar{x}$ in the interval $[-1, +1]$ was obtained from the user's original variable x in the interval $[x_{\min}, x_{\max}]$ by the linear transformation

$$\bar{x} = \frac{2x - (x_{\max} + x_{\min})}{x_{\max} - x_{\min}}$$

and that the user requires the integral to be with respect to the variable x. If the integral with respect to $\bar{x}$ is required, set $x_{\max} = 1$ and $x_{\min} = -1$.

Values of the integral can subsequently be computed, from the coefficients obtained, by using E02AKF.

The method employed is that of Chebyshev-series [1], Chapter 8, modified for integrating with respect to x. Initially taking $a_{n+1} = a_{n+2} = 0$, the routine forms successively

$$a'_i = \frac{a_{i-1} - a_{i+1}}{2i} \times \frac{x_{\max} - x_{\min}}{2}, \quad i = n+1, n, \ldots, 1.$$

The constant coefficient a'_0 is chosen so that $q(x)$ is equal to a specified value, QATM1, at the lower end-point of the interval on which it is defined, i.e., $\bar{x} = -1$, which corresponds to $x = x_{\min}$.

4 References

[1] (1961) Chebyshev-series *Modern Computing Methods, NPL Notes on Applied Science* **16** HMSO (2nd Edition)

5 Parameters

1: NP1 — INTEGER *Input*

On entry: $n+1$, where n is the degree of the given polynomial $p(x)$. Thus NP1 is the number of coefficients in this polynomial.

Constraint: NP1 ≥ 1.

2: XMIN — ***real*** *Input*

3: XMAX — ***real*** *Input*

On entry: the lower and upper end-points respectively of the interval $[x_{\min}, x_{\max}]$. The Chebyshev-series representation is in terms of the normalised variable $\bar{x}$, where

$$\bar{x} = \frac{2x-(x_{\max}+x_{\min})}{x_{\max}-x_{\min}}.$$

Constraint: XMAX > XMIN.

4: A(LA) — ***real*** array *Input*

On entry: the Chebyshev coefficients of the polynomial $p(x)$. Specifically, element $1+i\times$ IA1 of A must contain the coefficient a_i, for $i=0,1,\ldots,n$. Only these $n+1$ elements will be accessed.

Unchanged on exit, but see AINT, below.

5: IA1 — INTEGER *Input*

On entry: the index increment of A. Most frequently the Chebyshev coefficients are stored in adjacent elements of A, and IA1 must be set to 1. However, if for example, they are stored in A(1),A(4),A(7),..., then the value of IA1 must be 3. See also Section 8.

Constraint: IA1 ≥ 1.

6: LA — INTEGER *Input*

On entry: the dimension of the array A as declared in the (sub)program from which E02AJF is called.

Constraint: LA $\geq 1+($ NP1 $-1)\times$ IA1.

7: QATM1 — ***real*** *Input*

On entry: the value that the integrated polynomial is required to have at the lower end-point of its interval of definition, i.e., at $\bar{x}=-1$ which corresponds to $x=x_{\min}$. Thus, QATM1 is a constant of integration and will normally be set to zero by the user.

8: AINT(LAINT) — ***real*** array *Output*

On exit: the Chebyshev coefficients of the integral $q(x)$. (The integration is with respect to the variable x, and the constant coefficient is chosen so that $q(x_{\min})$ equals QATM1). Specifically, element $1+i\times$ IAINT1 of AINT contains the coefficient a_i', for $i=0,1,\ldots,n+1$. A call of the routine may have the array name AINT the same as A, provided that note is taken of the order in which elements are overwritten when choosing starting elements and increments IA1 and IAINT1: i.e., the coefficients, $a_0,a_1,\ldots,a_{i-2}$ must be intact after coefficient a_i' is stored. In particular it is possible to overwrite the a_i entirely by having IA1 = IAINT1, and the actual array for A and AINT identical.

9: IAINT1 — INTEGER *Input*

On entry: the index increment of AINT. Most frequently the Chebyshev coefficients are required in adjacent elements of AINT, and IAINT1 must be set to 1. However, if, for example, they are to be stored in AINT(1),AINT(4),AINT(7),..., then the value of IAINT1 must be 3. See also Section 8.

Constraint: IAINT1 ≥ 1.

10: LAINT — INTEGER *Input*

On entry: the dimension of the array AINT as declared in the (sub)program from which E02AJF is called.

Constraint: LAINT $\geq 1 + \text{NP1} \times \text{IAINT1}$.

11: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL $= 1$

On entry, NP1 < 1,
or XMAX $\leq$ XMIN,
or IA1 < 1,
or LA $\leq (\text{NP1} - 1) \times \text{IA1}$,
or IAINT1 < 1,
or LAINT $\leq \text{NP1} \times \text{IAINT1}$.

7 Accuracy

In general there is a gain in precision in numerical integration, in this case associated with the division by $2i$ in the formula quoted in Section 3.

8 Further Comments

The time taken by the routine is approximately proportional to $n + 1$.

The increments IA1, IAINT1 are included as parameters to give a degree of flexibility which, for example, allows a polynomial in two variables to be integrated with respect to either variable without rearranging the coefficients.

9 Example

Suppose a polynomial has been computed in Chebyshev-series form to fit data over the interval $[-0.5, 2.5]$. The following program evaluates the integral of the polynomial from 0.0 to 2.0. (For the purpose of this example, XMIN, XMAX and the Chebyshev coefficients are simply supplied in DATA statements. Normally a program would read in or generate data and compute the fitted polynomial).

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     E02AJF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NP1, LA, LAINT
      PARAMETER        (NP1=7,LA=NP1,LAINT=NP1+1)
      INTEGER          NOUT
```

```
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      real             RA, RB, RESULT, XA, XB, XMAX, XMIN
      INTEGER          I, IFAIL
*     .. Local Arrays ..
      real             A(LA), AINT(LAINT)
*     .. External Subroutines ..
      EXTERNAL         E02AJF, E02AKF
*     .. Data statements ..
      DATA             XMIN, XMAX/-0.5e0, 2.5e0/
      DATA             (A(I),I=1,NP1)/2.53213e0, 1.13032e0, 0.27150e0,
     +                 0.04434e0, 0.00547e0, 0.00054e0, 0.00004e0/
*     .. Executable Statements ..
      WRITE (NOUT,*) 'E02AJF Example Program Results'
      IFAIL = 0
*
      CALL E02AJF(NP1,XMIN,XMAX,A,1,LA,0.0e0,AINT,1,LAINT,IFAIL)
*
      XA = 0.0e0
      XB = 2.0e0
*
      CALL E02AKF(NP1+1,XMIN,XMAX,AINT,1,LAINT,XA,RA,IFAIL)
      CALL E02AKF(NP1+1,XMIN,XMAX,AINT,1,LAINT,XB,RB,IFAIL)
*
      RESULT = RB - RA
      WRITE (NOUT,*)
      WRITE (NOUT,99999) 'Value of definite integral is ', RESULT
      STOP
*
99999 FORMAT (1X,A,F10.4)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 E02AJF Example Program Results

 Value of definite integral is     2.1515
```